

# adacf: QCD DIS anomalous dimensions and coefficient functions

K.K.

July 23, 2006

## Abstract

Mathematica package `adacf.m` provides collection of QCD anomalous dimensions and Wilson coefficient functions for unpolarized<sup>1</sup> DIS up to NNLO order. This document presents slightly reformatted and documented source of this package, where it is specified from where the particular expressions are taken. Functions provided by the package are documented in the zeroth section. For installation instructions, see README file coming with the package.

## 0. Basic usage of package

(Information contained in this zeroth section is also available within *Mathematica* via commands `?Wgamma` and `?Wc`.)

### 0.1. Anomalous dimensions — `Wgamma[]`

- `Wgamma[{label}, {plr, OP}, n]` gives n-th Mellin moment of anomalous dimension (gamma) of operator OP i.e.  $^{OP}\gamma_n^{\text{plr}}$
- `Wgamma[{label}, {plr, OP}, p, n]` gives coefficient  $^{OP}\gamma_n^{\text{plr}(p)}$  of  $\alpha_s^p$  of this moment, up to a normalization specified by label.
  - `plr` can be one of `{V,A}` standing for unpolarized (vector) and polarized (axial-vector) case, respectively.
  - `OP` can be one of `{NSP, NSM, QQ, QG, GQ, GG}`. `NSP=NS(+)` and `NSM=NS(-)`<sup>2</sup>
  - Various conventions and results of various papers are specified by different `{label}`. At the moment, implemented labels are:

---

<sup>1</sup>Anomalous dimensions for the *polarized* DIS up to NLO are also available.

<sup>2</sup>For definition of  $\text{NS}^\pm$  see e.g. Eq. (2.2)-(2.3) of (1). For our DVCS calculations we need  $\text{NS}^{(+)}$ . By the way, for convenience, in the code variables `NSP` and `NSM` are predefined to have values 1 and -1, respectively.

- \* **as4p** - expansion in  $\alpha_s/(4\pi)$
  - \* **as2p** - expansion in  $\alpha_s/(2\pi)$
  - \* **RV** - results for integer moments from Retey and Vermaseren, hep-ph/0007294
- Also, by installing binary package **adacf.exe**, much faster functions obtained using Fortran are available using syntax: **Wgamma**[{**as2pF**}, **NF**, {**plr**, **OP**}, **p**, **n**], where number of active flavors **NF** needs to be specified, and other labels are as above.

## 0.2. Wilson coefficients — **Wc** []

- **Wc**[{**label**}, {**plr**, **f-fact**, **OP**}, **n**] gives n-th Mellin moment of Wilson coefficient function  $c$  of operator **OP**, appearing in expansion of form factor **f-fact**, i.e.  ${}^{OP}c_{f-fact,n}^{\text{plr}}$
- **Wc**[{**label**}, {**plr**, **f-fact**, **OP**}, **p**, **n**] gives coefficient  ${}^{OP}c_{f-fact,n}^{\text{plr}(p)}$  of  $\alpha_s^p$  of this moment, up to a normalization specified by label.
  - **plr** can be one of {**V**,**A**} standing for unpolarized (vector) and polarized (axial-vector) case, respectively.
  - **f-fact** can be one of {**F1**, **F2**, **F3**, **FL**}
  - **OP** can be one of {**NSP**, **NSM**, **Q**, **G**}. **NSP**=**NS**<sup>(+)</sup> and **NSM**=**NS**<sup>(-)</sup>
  - Various conventions and results of various papers are specified by different {**label**}. At the moment, implemented labels are:
    - \* **as4p** - expansion in  $\alpha_s/(4\pi)$
    - \* **as2p** - expansion in  $\alpha_s/(2\pi)$
    - \* **RV** - results for integer moments from Retey and Vermaseren, hep-ph/0007294
- Also, by installing binary package **adacf.exe**, much faster functions are available using syntax: **Wc**[{**as2pF**}, **NF**, {**plr**, **f-fact** **OP**}, **p**, **n**], where labels are as above, and number of active flavors **NF** needs to be specified.

## 1. Harmonic sums

Often occurring sums  $S'_k$  and  $\tilde{S}$  are implemented according to Eq. (A.7) of (2) and (30) from (3), respectively. The only non-trivial part is Mellin transform of  $Li_2(x)/(1+x)$  implemented according to approximation formula by (3) Eq. (33), which should be correct to accuracy of  $10^{-7}$ .

```
beta[x_] := (1/2)*(PolyGamma[0, (x + 1)/2] - PolyGamma[0, x/2])
```

```
(* Mellin transform of Li2(x)/(1+x) according to Bluemlein and Kurth, hep-ph/9708388 *)
```

```

aBK[1] = 0.9999964239
aBK[2] = -0.4998741238
aBK[3] = 0.3317990258
aBK[4] = -0.2407338084
aBK[5] = 0.1676540711
aBK[6] = -0.0953293897
aBK[7] = 0.0360884937
aBK[8] = -0.0064535442

MellinF2[{BK}, n_] := Zeta[2]*Log[2] -
  Sum[aBK[k]*((n - 1)*(Zeta[2]/(n + k - 1) - (PolyGamma[0, n + k] +
    EulerGamma)/(n + k - 1)^2) + (PolyGamma[0, n + k] + EulerGamma)/
    (n + k - 1)), {k, 1, 8}]

MellinF2[n_] := MellinF2[{BK}, n]

S[{a_}, z_] := HarmonicNumber[z, a]

S[parity_, {-2, 1}, n_] := (-(5/8))*Zeta[3] +
  parity*(S1[n]/n^2 - Zeta[2]*beta[n] + MellinF2[n])

S1[z_] := HarmonicNumber[z]
S2[z_] := HarmonicNumber[z, 2]
S3[z_] := HarmonicNumber[z, 3]
S4[z_] := HarmonicNumber[z, 4]

Spr[parity_, {a_}, N_] := (1/2)*(1 + parity)*S[{a}, N/2] +
  (1/2)*(1 - parity)*S[{a}, (N - 1)/2]

```

Note that here variable **parity** corresponds to the values from the following table

	V	A
NS <sup>+</sup>	+1	-1
NS <sup>-</sup>	-1	+1
singlet	+1	-1

## 2. Unpolarized - Anomalous dimensions

In following we use

$$C_F = \frac{N_c^2 - 1}{2N_c}, \quad C_A = N_c, \quad T_F = \frac{1}{2} \quad (1)$$

for  $SU(N_c)$  group constants.

## 2.1 Non-singlet

LO ( $\alpha_s^1$ ) -  $\gamma^{(0)}$

From everywhere, e.g. (4) Eq. (2.79a)

```
Wgamma[{as4p},{V, NS_ /; MemberQ[{NSP,NSM},NS]], 0, n_] := 2*CF*(-3 - 2/(n*(n + 1)) + 4*S1[n])
```

NLO ( $\alpha_s^2$ ) -  $\gamma^{(1)}$

From (5) Eq. (5.30)

```
Wgamma[{as4p},{V, NS_ /; MemberQ[{NSP,NSM},NS]], 1, n_] := (CF^2 - (1/2)*CF*CA)*
  (16*S1[n]*((2*n + 1)/(n^2*(n + 1)^2)) + 16*(2*S1[n] - 1/(n*(n + 1)))*
  (S2[n] - Spr[NS, {2}, n]) + 64*S[NS, {-2, 1}, n] + 24*S2[n] - 3 -
  8*Spr[NS, {3}, n] - 8*((3*n^3 + n^2 - 1)/(n^3*(n + 1)^3)) -
  16*NS*((2*n^2 + 2*n + 1)/(n^3*(n + 1)^3))) +
  CF*CA*(S1[n]*(536/9 + 8*((2*n + 1)/(n^2*(n + 1)^2))) - 16*S1[n]*S2[n] +
  S2[n]*(-(52/3) + 8/(n*(n + 1))) - 43/6 -
  4*((151*n^4 + 263*n^3 + 97*n^2 + 3*n + 9)/(9*n^3*(n + 1)^3))) +
  CF*NF*TF*((-(160/9))*S1[n] + (32/3)*S2[n] + 4/3 +
  16*((11*n^2 + 5*n - 3)/(9*n^2*(n + 1)^2)))
```

NNLO ( $\alpha_s^3$ ) -  $\gamma^{(2)}$

Approximate formula by (6), Eq (4.22). Mellin-transform to n-space by K.K, see notebook MellinTable.nb. Most of the expressions were derived using (7).

Formula should be accurate to one per mill. In the same paper exact formulas are given, but they are very complicated and very non-trivial to analytically continue to complex Mellin moments. See discussion below Eq. (4.24) of (6).

Only NS<sup>(+)</sup> implemented at the moment, see footnote 1.

SU(3) group constants values are substituted ( $C_F \rightarrow 4/3, \dots$ )

These comments apply everywhere below where it is specified that formulas are “approximate”.

```
Wgamma[{as4p},{V, NSP}, 2, n_] :=(
-2*(1295.384 + 1024/(27*n^5) - 1600/(9*n^4) + 589.8/n^3 - 1258/n^2 +
  1641.1/n - 3135/(1 + n) + 243.6/(2 + n) - 522.1/(3 + n) - (714.1*S1[n])/n -
  1174.898*(-1/n + S1[n]) + 563.9*(S1[n]/n^2 + (-Zeta[2] + S2[n])/n) +
  NF*(-173.927 + 128/(9*n^4) - 5216/(81*n^3) + 152.6/n^2 - 197./n +
  8.982/(1 + n)^4 + 381.1/(1 + n) + 72.94/(2 + n) +
  44.79/(3 + n) + (5120*S1[n])/(81*n) + 183.187*(-1/n + S1[n]) -
  56.66*(S1[n]/n^2 + (-Zeta[2] + S2[n])/n)) +
  NF^2*(-68/27 + 32/(27*n^3) - 352/(81*n^2) + 448/(81*n) -
  32/(27*(1 + n)^3) + 352/(81*(1 + n)^2) - 128/(27*(1 + n)) +
  (160*6*Zeta[2])/243 + (320*(-Zeta[2] + S2[n]))/81 +
  (64*(-n^(-1) + S1[n]))/81 - (64*(S3[n] - Zeta[3]))/27 -
  (64*Zeta[3])/27) + (513.6*(Zeta[2]/n - S1[n]/n^2 - S2[n]/n - S3[n] +
  Zeta[3]))/n )
```

## 2.2 Singlet

LO ( $\alpha_s^1$ ) -  $\gamma^{(0)}$

From everywhere, e.g. (4) Eq. (2.79)

```
Wgamma[{as4p},{V,QQ}, 0, n_] := Wgamma[{as4p},{V,NSP}, 0, n]
Wgamma[{as4p},{V,QQ}, 0, n_] := -8*TF*NF*(n^2 + n + 2)/(n*(n+1)*(n+2))
Wgamma[{as4p},{V,QQ}, 0, n_] := -4*CF*(n^2 + n + 2)/((n-1)*n*(n+1))
Wgamma[{as4p},{V,GG}, 0, n_] := (-8*CA*(-S1[n]+1/(n*(n-1))+1/((n+1)*(n+2))) +
8*TF*NF/3 - 2*11*CA/3)
```

NLO ( $\alpha_s^2$ ) -  $\gamma^{(1)}$

From (8), Appendix B

```
Wgamma[{as4p},{V,QQ}, 1, n_] := (Wgamma[{as4p},{V,NSP}, 1, n] -
16*CF*TF*NF*(5*n^5+32*n^4+49*n^3+38*n^2+28*n+8)/((n-1)*n^3*(n+1)^3*(n+2)^2))

Wgamma[{as4p},{V,QQ}, 1, n_] := (-1)*(8*CA*TF*NF*((-2*S1[n]^2 + 2*S2[n] - 2*Spr[1, {2}, n])*
((n^2 + n + 2)/(n*(n + 1)*(n + 2))) + (8*S1[n]*(2*n + 3))/
((n + 1)^2*(n + 2)^2) + 2*((n^9 + 6*n^8 + 15*n^7 + 25*n^6 + 36*n^5 +
85*n^4 + 128*n^3 + 104*n^2 + 64*n + 16)/((n - 1)*n^3*(n + 1)^3*
(n + 2)^3))) + 8*CF*TF*NF*((2*S1[n]^2 - 2*S2[n] + 5)*
((n^2 + n + 2)/(n*(n + 1)*(n + 2))) - (4*S1[n])/n^2 +
(11*n^4 + 26*n^3 + 15*n^2 + 8*n + 4)/(n^3*(n + 1)^3*(n + 2))))

Wgamma[{as4p},{V,GG}, 1, n_] := (-1)*(4*CF^2*((-2*S1[n]^2 + 10*S1[n] - 2*S2[n])*((n^2 + n + 2)/
((n - 1)*n*(n + 1))) - (4*S1[n])/((n + 1)^2 -
(12*n^6 + 30*n^5 + 43*n^4 + 28*n^3 - n^2 - 12*n - 4)/
((n - 1)*n^3*(n + 1)^3)) +
8*CF*CA*((S1[n]^2 + S2[n] - Spr[1, {2}, n])*((n^2 + n + 2)/
((n - 1)*n*(n + 1))) - S1[n]*((17*n^4 + 41*n^2 - 22*n - 12)/
(3*(n - 1)^2*n^2*(n + 1))) + (109*n^9 + 621*n^8 + 1400*n^7 +
1678*n^6 + 695*n^5 - 1031*n^4 - 1304*n^3 - 152*n^2 + 432*n + 144)/
(9*(n - 1)^2*n^3*(n + 1)^3*(n + 2)^2)) +
(32/3)*CF*NF*TF*((S1[n] - 8/3)*((n^2 + n + 2)/((n - 1)*n*(n + 1))) +
1/(n + 1)^2))

Wgamma[{as4p},{V,GG}, 1, n_] := CA*NF*TF*((-160/9)*S1[n] + 32/3 +
(16/9)*((38*n^4 + 76*n^3 + 94*n^2 + 56*n + 12)/((n - 1)*n^2*(n + 1)^2*
(n + 2)))) + CF*NF*TF*
(8 + 16*((2*n^6 + 4*n^5 + n^4 - 10*n^3 - 5*n^2 - 4*n - 4)/
((n - 1)*n^3*(n + 1)^3*(n + 2)))) +
CA^2*((536/9)*S1[n] + 64*S1[n]*((2*n^5 + 5*n^4 + 8*n^3 + 7*n^2 - 2*n -
2)/((n - 1)^2*n^2*(n + 1)^2*(n + 2)^2)) - 64/3 +
32*Spr[1, {2}, n]*((n^2 + n + 1)/((n - 1)*n*(n + 1)*(n + 2))) -
16*S1[n]*Spr[1, {2}, n] + 32*S[1, {-2, 1}, n] - 4*Spr[1, {3}, n] -
(4/9)*((457*n^9 + 2742*n^8 + 6040*n^7 + 6098*n^6 + 1567*n^5 -
2344*n^4 - 1632*n^3 + 560*n^2 + 1488*n + 576)/
((n - 1)^2*n^3*(n + 1)^3*(n + 2)^3)))
```

NNLO ( $\alpha_s^3$ ) -  $\gamma^{(2)}$

Approximate SU(3) formulas (4.32 – 4.35) from (9). Mellin-transform to n-space by K.K.

```
ps2Naux[n_] := (
NF^2*(256/(81*(-1 + n)) - 64/(9*n^4) + 35.78/n^3 - 61.75/n^2 + 100.1/n -
125.2/(1 + n) + 49.26/(2 + n) - 12.59/(3 + n) - (5.944*S1[n])/n +
```

```

(1.778*(S1[n]^2 + S2[n]))/n - 1.889*(S1[n]/n^2 + (-Pi^2/6 + S2[n])/n)) +
NF*(3584/(27*(-1 + n)^2) - 506./(-1 + n) + 1280/(9*n^5) + 800/(3*n^4) +
262.8/n^3 + 661.6/n^2 + 177.4/n + 392.9/(1 + n) - 101.4/(2 + n) +
(72.11*S1[n])/n - (9.751*(S1[n]^2 + S2[n]))/n -
57.04*(S1[n]/n^2 + (-Pi^2/6 + S2[n])/n) +
(5.926*(S1[n]^3 + 3*S1[n]*S2[n] + 2*S3[n]))/n)
)

(* (-1) is usual convention, factor 2 is Vermaseren w.r.t rest of the world
and [n]-[n+1] corresponds to multiplication by (1-x) in x-space: *)
Wgamma[{as4p},{V,PS}, 2, n_] := - 2*(ps2Naux[n] - ps2Naux[n + 1])

Wgamma[{as4p},{V,QQ}, 2, n_] := (
Wgamma[{as4p}, {V,NSP}, 2, n] + Wgamma[{as4p}, {V,PS}, 2, n] )

Wgamma[{as4p},{V,QG}, 2, n_] := -2*(
NF^2*(1112/(243*(-1 + n)) - 128/(3*n^5) + 752/(9*n^4) - 181.6/n^3 +
254./n^2 - 252./n - 70.19999999999999/(1 + n)^4 - 196.14/(1 + n)^3 +
158./(1 + n) + 145.4/(2 + n) - 139.28/(3 + n) + (5.496*S1[n])/n +
(200*(S1[n]^2 + S2[n]))/(27*n) - 53.09*(S1[n]/n^2 + (-Pi^2/6 + S2[n])/n) -
(20*(S1[n]^3 + 3*S1[n]*S2[n] + 2*S3[n]))/(27*n) -
(161.232*(Pi^2/(6*n) - S1[n]/n^2 - S2[n]/n - S3[n] + Zeta[3]))/n) +
NF*(896/(3*(-1 + n)^2) - 1268.3/(-1 + n) + 4288/(9*n^5) + 88/n^4 +
1763./n^3 - 424.9/n^2 + 2522/n + 1515./(1 + n)^4 - 3316/(1 + n) +
2126/(2 + n) - (104.42*S1[n])/n - (120.5*(S1[n]^2 + S2[n]))/n +
1823*(S1[n]/n^2 + (-Pi^2/6 + S2[n])/n) +
(70*(S1[n]^3 + 3*S1[n]*S2[n] + 2*S3[n]))/(9*n) +
(100*(S1[n]^4/n + (6*S1[n]^2*S2[n])/n + (3*S2[n]^2)/n +
(8*S1[n]*S3[n])/n + (6*S4[n])/n))/27 -
(50.44*(Pi^2/(6*n) - S1[n]/n^2 - S2[n]/n - S3[n] + Zeta[3]))/n)
)

Wgamma[{as4p},{V,GQ}, 2, n_] := -2*(
-1189.3/(-1 + n)^2 + 6163.1/(-1 + n) - 34304/(27*n^5) - 3136/(3*n^4) -
3588/n^3 - 4033/n^2 - 4307/n - 1945.8/(1 + n)^3 + 489.3/(1 + n) +
1452/(2 + n) + 146./(3 + n) - (2193*S1[n])/n + (606.3*(S1[n]^2 + S2[n]))/n +
NF^2*((-64*(-1 + n)^(-1) - n^(-1) - 2/(1 + n)))/27 +
(320*(-(S1[-1 + n])/(-1 + n)) + S1[n]/n - (4*S1[1 + n])/(5*(1 + n)))/27 +
(32*((S1[-1 + n]^2 + S2[-1 + n])/(-1 + n) - (S1[n]^2 + S2[n])/n +
(S1[1 + n]^2 + S2[1 + n])/(2*(1 + n))))/9 -
(2200*(S1[n]^3 + 3*S1[n]*S2[n] + 2*S3[n]))/(27*n) +
NF*(-71.082/(-1 + n)^2 - 46.41/(-1 + n) + 1024/(9*n^5) - 1408/(27*n^4) +
40.78/n^3 - 174.8/n^2 - 183.8/n + 217.2/(1 + n)^3 + 33.35/(1 + n) -
277.9/(2 + n) + (296.7*S1[n])/n - (68.069*(S1[n]^2 + S2[n]))/n -
49.68*(S1[n]/n^2 + (-Pi^2/6 + S2[n])/n) +
(400*(S1[n]^3 + 3*S1[n]*S2[n] + 2*S3[n]))/(81*n) +
(400*(S1[n]^4/n + (6*S1[n]^2*S2[n])/n + (3*S2[n]^2)/n + (8*S1[n]*S3[n])/n +
(6*S4[n])/n))/81 - (894.6*(Pi^2/(6*n) - S1[n]/n^2 - S2[n]/n - S3[n] +
Zeta[3]))/n
)

Wgamma[{as4p},{V,GG}, 2, n_] := -2*(
4425.894 - 2675.8/(-1 + n)^2 + 14214/(-1 + n) - 3456/n^5 - 432/n^4 -
14942/n^3 - 274.4/n^2 - 20852/n + 3968/(1 + n) - 3363/(2 + n) +
4848/(3 + n) - 2643.521*S1[-1 + n] - (3589*S1[n])/n +
7305*(S1[n]/n^2 + (-Pi^2/6 + S2[n])/n) +
(17514*(Pi^2/(6*n) - S1[n]/n^2 - S2[n]/n - S3[n] + Zeta[3]))/n +
NF*(-528.723 - 157.27/(-1 + n)^2 + 182.96/(-1 + n) + 4096/(9*n^5) -
1664/(3*n^4) + 982.6/n^3 - 1541/n^2 - 350.2/n + 755.7/(1 + n) -
713.8/(2 + n) + 559.3/(3 + n) + 412.172*S1[-1 + n] + (320*S1[n])/n +
26.15*(S1[n]/n^2 + (-Pi^2/6 + S2[n])/n) -
(1617.4*(Pi^2/(6*n) - S1[n]/n^2 - S2[n]/n - S3[n] + Zeta[3]))/n +
NF^2*(6.463 - 680/(243*(-1 + n)) + 64/(9*n^4) + 19.36/n^3 + 3.422/n^2 -
13.878/n + 153.4/(1 + n) - 187.7/(2 + n) + 52.75/(3 + n) +

```

```

(16*S1[-1 + n])/9 - 115.6*(S1[n]/n^2 + (-Pi^2/6 + S2[n])/n) +
85.25*(S1[1 + n]/(1 + n)^2 + (-Pi^2/6 + S2[1 + n])/(1 + n)) -
(126.46*(Pi^2/(6*n) - S1[n]/n^2 - S2[n]/n - S3[n] + Zeta[3]))/n)
)

```

By the way, the symbol PS stands for “pure singlet”.

### 3. Unpolarized - Wilson coefficient functions

LO ( $\alpha_s^0$ ) -  $c^{(0)}$  - both singlet and non-singlet

```

Wc[{as4p},
{V,ff_ /; MemberQ[{F1,F2,F3}, ff], op_ /; MemberQ[{Q,NSP,NSM}, op]}, 0, n_] := 1
Wc[{as4p}, {V,ff_,op_}, 0, n_] := 0

```

#### 3.1 Non-singlet

NLO ( $\alpha_s^1$ ) -  $c^{(1)}$

From everywhere, e.g. (10) Eqs. (3.7-9) (where last term should be removed in  $\overline{MS}$  scheme, and  $S_{1,1} = (S_1^2 + S_2)/2$ )

```

Wc[{as4p}, {V,F2,NS_ /; MemberQ[{NSP,NSM},NS]}, 1, n_] := CF*(
2*S1[n]^2 - 2*S2[n] + (3 - 2/(n*(n + 1)))*S1[n] + 2/n^2 + 3/n + 4/(n + 1) - 9)

```

```

Wc[{as4p}, {V,FL,NS_ /; MemberQ[{NSP,NSM},NS]}, 1, n_] := CF*(4/(n + 1))

```

There is a sign disagreement in definition of gluon operator between (10; 4) and most of the rest of the literature (e.g. (11)). However, they all agree on the sign of the corresponding Wilson coefficients, so I guess that “rest of the literature” should be trusted on the definition of the gluon operator being

$$O_G^{\mu_1 \dots \mu_n} = \frac{1}{2} G_a^{\mu_1 \nu} (iD^{\mu_2}) \dots (iD^{\mu_{n-1}}) G_{a\nu}^{\mu_n} \Big|_{\text{symmetric, traceless}}$$

NNLO ( $\alpha_s^2$ ) -  $c^{(2)}$

From (1), Appendix, approximate formula for SU(3).

```

Wc[{as4p}, {V,F2,NSP}, 2, N_] := 3.55555*(6*S4[N] + 8*S3[N]*S1[N] + 3*S2[N]^2 +
6*S2[N]*S1[N]^2 + S1[N]^4) + 20.4444*(2*S3[N] + 3*S2[N]*S1[N] +
S1[N]^3) - 15.5525*(S2[N] + S1[N]^2) - 188.64*S1[N] +
(165.356/N)*S3[N] - (15.38/N^2 - 9.7467/N)*S2[N] +
(358.503/N)*S2[N]*S1[N] + (2.9678/N)*S1[N]^3 +
(174.8/N^2 + 9.7467/N)*S1[N]^2 - (190.18/N^3 - 116.734/N)*S1[N] +
17.01/N^4 - 34.16/N^3 + 306.849/N^2 - 72.5824/N - 1008/(N + 1) -
338.044 + NF*(-0.59259*(2*S3[N] + 3*S2[N]*S1[N] + S1[N]^3) -
4.2963*(S2[N] + S1[N]^2) - 6.3489*S1[N] - (6.072/N)*S3[N] -
(6.072/N^2 - 18.0408/N)*S2[N] - (6.072/N^3 - 17.97/N^2 + 14.3574/N)*
S1[N] + (0.07078/N)*S1[N]^2 + 4.488/N^3 + 4.21808/N^2 - 21.6028/N -
37.91/(N + 1) + 46.8406)

```

```

Wc[{as4p}, {V,FL,NSP}, 2, N_] := -((136.88/N)*S2[N]) + (13.62/N)*S1[N]^2 +
(55.79/N - 150.5/N^2)*S1[N] - 0.062/N^3 + 14.85/N^2 + 207.153/N +
53.12/(N + 1)^3 + 97.48/(N + 1) - 0.164 +
NF*(16/27)*(-(6/(N + 1))*S1[N]) + 6/N + 6/(N + 1)^2 - 25/(N + 1))

```

## 3.2 Singlet

NLO ( $\alpha_s^1$ ) -  $c^{(1)}$

From (10)

```
(* Eq. (6.18-19) *)
Wc[{as4p}, {V,F2,Q}, 1, n_] := Wc[{as4p}, {V,F2,NSP}, 1, n]
Wc[{as4p}, {V,FL,Q}, 1, n_] := Wc[{as4p}, {V,FL,NSP}, 1, n]

(* Eq. (6.24-25) *)
Wc[{as4p}, {V,F2,G}, 1, n_] := 2*NF*( 4/(n+1) - 4/(n+2) + 1/n^2 -
  (n^2+n+2)*(1+S1[n])/(n*(n+1)*(n+2)) )
Wc[{as4p}, {V,FL,G}, 1, n_] := 8*NF/(n+1)/(n+2)
```

NNLO ( $\alpha_s^2$ ) -  $c^{(2)}$

From (12), Appendix. Approximate formula for SU(3).

Note that there are some typos/errors in Eqs. (A.3) and (A.4).

```
Wc[{as4p}, {V,F2,PS}, 2, N_] := NF*(
  (((49.702)/(N)) - ((27.802)/(N+1)) )*S3[N]
  + (((49.5)/(N^2)) + ((30.23)/(N))
    - ((27.903)/((N+1)^2)) )*S2[N]
  + 0.101 ( ((1)/(N)) - ((1)/(N+1)) )
  (3*S2[N]*S1[N] + S1[N]^3 )
  (* Next row is missing from the preprint Eq. (A.3) ! *)
  -0.303*S1[N]^2/(N+1)^2
  + ( ((49.5)/(N^3)) + ((30.23)/(N^2))
    - ((28.206)/((N+1)^3)) ) S1[N]
  + ((5.290)/(N-1)) - ((25.86)/(N^4))
  - ((4.172)/(N^3)) - ((121.205)/(N^2)) - ((114.519)/(N))
  - ((83.406)/((N+1)^4))
  + ((45.4003)/((N+1)^2)) + ((33.1769)/(N+1)) )

Wc[{as4p}, {V,FL,PS}, 2, N_] := NF*(
  ( - ((15.94)/(N)) + ((37.092)/(N+1))
    - ((26.364)/(N+2)) + ((5.212)/(N+3)) )*S1[N]
  - ((2.370)/(N-1))
  + ((0.842)/(N^3)) - ((28.09)/(N^2)) + ((26.38)/(N))
  + ((3.040)/((N+1)^3)) + ((65.182)/((N+1)^2))
  - ((88.678)/(N+1))
  - ((26.364)/((N+2)^2)) + ((91.756)/(N+2))
  + ((5.212)/((N+3)^2)) - ((27.088)/(N+3))
  )

Wc[{as4p}, {V,F2,Q}, 2, N_] := Wc[{as4p}, {V,F2,NSP}, 2, N]+Wc[{as4p}, {V,F2,PS}, 2, N]
Wc[{as4p}, {V,FL,Q}, 2, N_] := Wc[{as4p}, {V,FL,NSP}, 2, N]+Wc[{as4p}, {V,FL,PS}, 2, N]

Wc[{as4p}, {V,FL,G}, 2, N_] := NF*(
  ( ((94.74)/(N)) + ((1017.06)/(N+1)) + ((49.20)/(N+2)) )*S2[N]
  + ( ((94.74)/(N)) - ((143.94)/(N+1)) + ((49.20)/(N+2)) )*S1[N]^2
  + ( -((864.8)/(N)) + ((873.12)/((N+1)^2))
    + ((963.2)/(N+1))
    + ((98.40)/((N+2)^2)) - ((98.40)/(N+2)) )*S1[N]
  - ((5.333)/(N-1))
  - ((39.66)/(N^2)) + ((5.333)/(N))
  + ((2154.24)/((N+1)^3)) + ((1002.86)/((N+1)^2))
  - ((1909.768)/(N+1))
  (* - ((98.40)/((N+2)^3)) + ((98.40)/((N+2)^2)) Typo in preprint Eq. (A.4) !! *)
  + ((98.40)/((N+2)^3)) - ((98.40)/((N+2)^2))
```



)

```
Wc[{as4p}, {V,F2,G}, 2, N_] := NF*(
  ((2760.11)/(N)) + ((418.8)/(N+1)) )*S3[N]
+ ( ((2320)/(N^2)) - ((24)/(N)) + ((628.2)/((N+1)^2))) *S2[N]
+ ( ((1096.07)/(N)) + ((628.2)/(N+1)) ) *S2[N]*S1[N]
+ ( ((871.8)/(N^2)) - ((24)/(N)) + ((628.2)/((N+1)^2))) *S1[N]^2
- ( ((1494)/(N-1)) - ((1448.20)/(N^3))
+ ((1385.12)/(N)) - ((1256.4)/((N+1)^3)) ) *S1[N]
- ( ((215.845)/(N)) - ((209.4)/(N+1)) ) *S1[N]^3
+ ((1505.9)/(N-1)) - ((31.914)/(N^4))
- ((118.96)/(N^3)) - ((2097.4)/(N^2)) - ((4938.34)/(N))
+ ((1256.4)/(N+1)^4) - 0.271 )
```

## 4. Polarized - Anomalous dimensions

### 4.1 Non-singlet

LO ( $\alpha_s^1$ ) -  $\gamma^{(0)}$

Same as unpolarized.

```
Wgamma[{as4p}, {A, NS_ /; MemberQ[{NSP,NSM},NS]}, 0, n_] := Wgamma[{as4p}, {V, NS}, 0, n]
```

NLO ( $\alpha_s^2$ ) -  $\gamma^{(1)}$

Same as unpolarized, up to exchange  $NS^{(+)} \leftrightarrow NS^{(-)}$  according to discussion below Eq. (2.10) of (2)

```
Wgamma[{as4p}, {A, NS_ /; MemberQ[{NSP,NSM},NS]}, 1, n_] := Wgamma[{as4p}, {V, -NS}, 1, n]
```

### 4.2 Singlet

LO ( $\alpha_s^1$ ) -  $\gamma^{(0)}$

From (2), Appendix, Eq. (A.1).

```
Wgamma[{as4p}, {A, QQ}, 0, n_] := Wgamma[{as4p}, {V, NSM}, 0, n]
Wgamma[{as4p}, {A, QG}, 0, n_] := -8*TF*NF*(n-1)/(n*(n+1))
Wgamma[{as4p}, {A, GQ}, 0, n_] := -4*CF*(n+2)/(n*(n+1))
Wgamma[{as4p}, {A, GG}, 0, n_] := 4*CA*(2*S1[n]-4/(n*(n+1))-11/6) + 8*TF*NF/3
```

NLO ( $\alpha_s^2$ ) -  $\gamma^{(1)}$

From (2), Appendix, Eq. (A.2-6).

```
Wgamma[{as4p}, {A, QQ}, 1, n_] := ( Wgamma[{as4p}, {V, NSM}, 1, n] +
  16*CF*TF*NF*(n^4+2*n^3 + 2*n^2 + 5*n + 2)/(n^3*(n+1)^3) )
Wgamma[{as4p}, {A, QG}, 1, n_] := ( 8*CF*TF*NF*(2*(n-1)/(n*(n+1))*
  (S2[n]-S1[n]^2) + 4*(n-1)/(n^2*(n+1))*S1[n] -
  (5*n^5+5*n^4-10*n^3-n^2 +3*n-2)/(n^3*(n+1)^3)) +
  16*CA*TF*NF*((n-1)/(n*(n+1)))*(-S2[n]+Spr[-1, {2}, n]+S1[n]^2) -
```

```

4/(n*(n+1)^2)*S1[n] - (n^5+n^4-4*n^3+3*n^2-7*n-2)/(n^3*(n+1)^3)) )

Wgamma[{as4p},{A,GQ}, 1, n_] := ( 32*CF*TF*NF*(-(n+2)/(3*n*(n+1))*S1[n] +
(5*n^2+12*n+4)/(9*n*(n+1)^2)) +
4*CF^2*(2*(n+2)/(n*(n+1))*(S2[n]+S1[n]^2)- 2*(3*n^2+7*n+2)/(n*(n+1)^2)*S1[n] +
(9*n^5+30*n^4+24*n^3-7*n^2-16*n-4)/(n^3*(n+1)^3)) +
8*CA*CF*((n+2)/(n*(n+1))*(-S2[n]+Spr[-1, {2}, n]-S1[n]^2)+
(11*n^2+22*n+12)/(3*n^2*(n+1))*S1[n] -
(76*n^5+271*n^4+254*n^3+41*n^2+72*n+36)/(9*n^3*(n+1)^3)) ) )

Wgamma[{as4p},{A,GG}, 1, n_] := (
8*CF*TF*NF*(n^6+3*n^5+5*n^4+n^3-8*n^2+2*n+4)/(n^3*(n+1)^3) +
32*CA*TF*NF*(-5/9*S1[n] + (3*n^4+6*n^3+16*n^2+13*n-3)/(9*n^2*(n+1)^2)) +
4*CA^2*(-Spr[-1, {3}, n]-4*S1[n]*Spr[-1, {2}, n] + 8*S[-1, {-2,1}, n] +
8/(n*(n+1))*Spr[-1, {2}, n] +
2*(67*n^4+134*n^3+67*n^2+144*n+72)/(9*n^2*(n+1)^2)*S1[n] -
(48*n^6+144*n^5+469*n^4+698*n^3+7*n^2+258*n+144)/(9*n^3*(n+1)^3)) )

```

## 5. Polarized - Wilson coefficient functions

NOT IMPLEMENTED YET.

This can be implemented using (13).

## 6. Combinations

Here are defined summed expressions LO+NLO+NNLO, as well as label {as2p} corresponding to expansion in  $\alpha_s/(2\pi)$

```

(* 2xF_1 = F_2 - F_L (but 2x is absorbed into definitions) *)
Wc[{as4p},{plr_,F1,op_}, ord_, n_] := (
Wc[{as4p},{plr,F2,op}, ord, n] - Wc[{as4p},{plr,FL,op}, ord, n] )

(* Combining: gamma = a_s gamma^(0) + a_s^2 \gamma^(1) + a_s^3 \gamma^(2) *)
Wgamma[{as4p},{ops_}, n_] := (
as4p*Wgamma[{as4p},{ops}, 0, n] + as4p^2*Wgamma[{as4p},{ops},1,n] +
as4p^3*Wgamma[{as4p},{ops},2,n])

Wc[{as4p},{ops_}, n_] := ( Wc[{as4p},{ops},0,n] +
as4p*Wc[{as4p},{ops}, 1, n] + as4p^2*Wc[{as4p},{ops},2,n] )

(* Defining label as2p - expansion in a_strong/(2 pi) *)
Wgamma[{as2p},{ops_}, ord_, n_] := Wgamma[{as4p},{ops}, ord, n]/2^(ord+1)
Wc[{as2p},{ops_}, ord_, n_] := Wc[{as4p},{ops}, ord, n]/2^ord

Wgamma[{as2p},{ops_}, n_] := (
as2p*Wgamma[{as2p},{ops}, 0, n] + as2p^2*Wgamma[{as2p},{ops},1,n] +
as2p^3*Wgamma[{as2p},{ops},2,n])

Wc[{as2p},{ops_}, n_] := ( Wc[{as2p},{ops},0,n] +
as2p*Wc[{as2p},{ops}, 1, n] + as2p^2*Wc[{as2p},{ops},2,n] )

```

## 7. Retey and Vermaseren

Below are all expressions from Section 5 of (11). Obtained by semi-automatic edit of original LaTeX source This is very convenient for checking other expressions

(\* Anomalous dimensions for spin-even operators contributing to F2 and FL \*)

```

WgammaRV[{RV,QQ}, 2] = (
3.555555556*as4p + as4p^2*( 48.32921811 - 3.160493827*NF - 1.975308642*f102*NF )
+as4p^3*( 859.4478372 - 133.4381617*NF - 1.229080933*NF*NF
+f102*( -42.21182429*NF - 3.445816187*NF*NF ) )
)

WgammaRV[{RV,QQ}, 4] = (
6.977777778*as4p + as4p^2*( 86.28665021 - 6.553580247*NF - 0.1060246914*f102*NF )
+as4p^3*( 1515.562363 - 244.728592*NF - 2.108515775*NF*NF
+f102*( -5.17013312*NF - 0.6789278464*NF*NF ) )
)

WgammaRV[{RV,QQ}, 6] = (
9.003174603*as4p + as4p^2*( 108.0184697 - 8.62925674*NF - 0.02080408883*f102*NF )
+as4p^3*( 1891.827779 - 307.4236889*NF - 2.570638992*NF*NF
+f102*( -2.526091192*NF - 0.2884556035*NF*NF ) )
)

WgammaRV[{RV,QQ}, 8] = (
10.45820106*as4p + as4p^2*( 123.7764525 - 10.14583662*NF - 0.006586485074*f102*NF )
+as4p^3*( 2164.091836 - 352.3116596*NF - 2.882493484*NF*NF
+f102*( -1.682156519*NF - 0.1620816452*NF*NF ) )
)

WgammaRV[{RV,QQ}, 10] = (
11.5969216*as4p + as4p^2*( 136.2741775 - 11.34594534*NF - 0.00269944007*f102*NF )
+as4p^3*( 2379.919952 - 387.6422968*NF - 3.115523145*NF*NF
+f102*( -1.256562245*NF - 0.1054295071*NF*NF ) )
)

WgammaRV[{RV,QQ}, 12] = (
12.53336293*as4p + as4p^2*( 146.6771964 - 12.34063169*NF - 0.001302012432*f102*NF )
+as4p^3*( 2559.641948 - 416.9091301*NF - 3.300349343*NF*NF
+f102*( -0.9957297081*NF - 0.07498848634*NF*NF ) )
)

WgammaRV[{RV,NS}, 14] = (
13.32896733*as4p + as4p^2*( 155.6071962 - 13.19066078*NF )
+as4p^3*( 2714.031720 - 441.9494048*NF - 3.452881831*NF*NF )
)

WgammaRV[{RV,QG}, 2] = (
-0.6666666667*as4p*NF - 7.543209877*as4p^2*NF
+as4p^3*( -37.62337275*NF + 12.11248285*NF*NF )
)

WgammaRV[{RV,QG}, 4] = (
-0.3666666667*as4p*NF + 1.290703704*as4p^2*NF
+as4p^3*( 33.58149273*NF + 6.06027262*NF*NF )
)

WgammaRV[{RV,QG}, 6] = (
-0.2619047619*as4p*NF + 2.761104812*as4p^2*NF
+as4p^3*( 33.41602135*NF + 3.537682102*NF*NF )
)

```

```

WgammaRV[{RV,QQ}, 8] = (
-0.20555555556*as4p*Nf + 3.243957223*as4p^2*Nf
+as4p^3*( 28.7612615*Nf + 2.225433112*Nf*Nf )
)

WgammaRV[{RV,QQ}, 10] = (
-0.1696969697*as4p*Nf + 3.407168695*as4p^2*Nf
+as4p^3*( 23.93704198*Nf + 1.449828678*Nf*Nf )
)

WgammaRV[{RV,QQ}, 12] = (
-0.1446886447*as4p*Nf + 3.438705999*as4p^2*Nf
+as4p^3*( 19.63230379*Nf + 0.9524545446*Nf*Nf )
)

WgammaRV[{RV,GQ}, 2] = (
-3.555555556*as4p + as4p^2*( -48.32921811 + 5.135802469*Nf )
+as4p^3*( -859.4478372 + 175.649986*Nf + 4.674897119*Nf*Nf )
)

WgammaRV[{RV,GQ}, 4] = (
-0.9777777778*as4p + as4p^2*( -16.1752428 + 0.6182716049*Nf )
+as4p^3*( -315.276255 + 39.82571027*Nf + 1.801843621*Nf*Nf )
)

WgammaRV[{RV,GQ}, 6] = (
-0.5587301587*as4p + as4p^2*( -9.496317796 + 0.08884857647*Nf )
+as4p^3*( -188.9088124 + 19.67944546*Nf + 1.087843741*Nf*Nf )
)

WgammaRV[{RV,GQ}, 8] = (
-0.3915343915*as4p + as4p^2*( -6.757603506 - 0.07061952353*Nf )
+as4p^3*( -134.7055042 + 12.3754454*Nf + 0.7536013741*Nf*Nf )
)

WgammaRV[{RV,GQ}, 10] = (
-0.3016835017*as4p + as4p^2*( -5.297576945 - 0.1348941718*Nf )
+as4p^3*( -104.911278 + 8.796702078*Nf + 0.5579674847*Nf*Nf )
)

WgammaRV[{RV,GQ}, 12] = (
-0.2455322455*as4p + as4p^2*( -4.398625917 - 0.1639529655*Nf )
+as4p^3*( -86.18107998 + 6.735609285*Nf + 0.4293379075*Nf*Nf )
)

WgammaRV[{RV,GG}, 2] = (
0.666666667*as4p*Nf + 7.543209877*as4p^2*Nf
+as4p^3*( 37.62337275*Nf - 12.11248285*Nf*Nf )
)

WgammaRV[{RV,GG}, 4] = (
as4p^2*( 128.178 - 13.64948148*Nf ) + as4p*( 12.6 + 0.666666667*Nf )
+as4p^3*( 2066.19278 - 401.3127939*Nf - 10.43150645*Nf*Nf )
)

WgammaRV[{RV,GG}, 6] = (
as4p^2*( 183.0538144 - 20.46668466*Nf ) + as4p*( 17.78571429 + 0.666666667*Nf )
+as4p^3*( 2987.042058 - 566.6373298*Nf - 10.78060861*Nf*Nf )
)

WgammaRV[{RV,GG}, 8] = (
as4p^2*( 219.6240988 - 24.69926432*Nf ) + as4p*( 21.26666667 + 0.666666667*Nf )
+as4p^3*( 3609.35419 - 673.9430658*Nf - 11.20133837*Nf*Nf )
)

```

```

WgammaRV[{RV,GG}, 10] = (
as4p^2*( 247.6655484 - 27.82178573*NF ) + as4p*( 23.92337662 + 0.6666666667*NF )
+as4p^3*( 4089.236943 - 755.1340541*NF - 11.57068198*NF*NF )
)

WgammaRV[{RV,GG}, 12] = (
as4p^2*( 270.6428892 - 30.31377688*NF ) + as4p*( 26.08168498 + 0.6666666667*NF )
+as4p^3*( 4483.563048 - 821.1236576*NF - 11.88665683*NF*NF )
)

(* The corresponding coefficient functions *)

WCRV[{RV,Q,2}, 2] = (
1 + 0.4444444444*as4p + as4p^2*( 17.69376589 - 5.333333333*NF - 2.189300412*f102*NF )
+as4p^3*( 442.7409693 - 165.1971095*NF - 24.09201335*f111*NF + 6.030272415*NF*NF
+f102*( -79.04486142*NF + 3.325504478*NF*NF ) )
)

WCRV[{RV,Q,2}, 4] = (
1 + 6.066666667*as4p + as4p^2*( 142.3434719 - 16.98791358*NF + 0.4858308642*f102*NF )
+as4p^3*( 4169.267888 - 901.2351626*NF - 18.21884618*f111*NF + 23.35503924*NF*NF
+f102*( 16.64834849*NF - 2.208630689*NF*NF ) )
)

WCRV[{RV,Q,2}, 6] = (
1 + 11.17671958*as4p + as4p^2*( 302.398735 - 28.0130504*NF + 0.4868787285*f102*NF )
+as4p^3*( 10069.63085 - 1816.322929*NF - 16.14271761*f111*NF + 42.66273116*NF*NF
+f102*( 24.11778813*NF - 1.525489143*NF*NF ) )
)

WCRV[{RV,Q,2}, 8] = (
1 + 15.52989418*as4p + as4p^2*( 470.807419 - 37.9248228*NF + 0.3859585393*f102*NF )
+as4p^3*( 17162.37245 - 2787.297692*NF - 15.09203827*f111*NF + 61.91177997*NF*NF
+f102*( 22.33201938*NF - 1.036308122*NF*NF ) )
)

WCRV[{RV,Q,2}, 10] = (
1 + 19.30061568*as4p + as4p^2*( 639.210663 - 46.86131842*NF + 0.3045901308*f102*NF )
+as4p^3*( 24953.13497 - 3770.10212*NF - 14.45874451*f111*NF + 80.52097973*NF*NF
+f102*( 19.53359559*NF - 0.7372434464*NF*NF ) )
)

WCRV[{RV,Q,2}, 12] = (
1 + 22.62841097*as4p + as4p^2*( 804.5854321 - 54.99446579*NF + 0.2451231747*f102*NF )
+as4p^3*( 33171.45501 - 4746.440949*NF - 14.03541028*f111*NF + 98.3483124*NF*NF
+f102*( 16.98652635*NF - 0.5471547625*NF*NF ) )
)

WCRV[{RV,NS,2}, 14] = (
(* RV have typo: 1 + 2.561093284*as4p + as4p^2*( 965.8132564 - 62.46549093*NF ) *)
1 + 25.61093284*as4p + as4p^2*( 965.8132564 - 62.46549093*NF )
+as4p^3*( 41657.11568 - 5708.215623*NF - 13.73240102*f111*NF + 115.3919490*NF*NF )
)

WCRV[{RV,G,2}, 2] = (
-0.5*as4p*NF - 8.918338961*as4p^2*NF
+as4p^3*( -130.7340963*NF + 29.37933515*NF*NF - 0.9007972776*f1g11*NF*NF )
)

WCRV[{RV,G,2}, 4] = (
-0.7388888889*as4p*NF - 14.27158692*as4p^2*NF
+as4p^3*( -346.4612756*NF + 46.52017564*NF*NF - 1.611816512*f1g11*NF*NF )
)

```

```

WCRV[{RV,G,2} ,6] = (
-0.7051587302*as4p*Nf - 20.06849828*as4p^2*Nf
+as4p^3*( -715.0372438*Nf + 61.28545096*Nf*Nf - 1.496036938*f1g11*Nf*Nf )
)

WCRV[{RV,G,2} ,8] = (
-0.6440873016*as4p*Nf - 23.17873524*as4p^2*Nf
+as4p^3*( -996.5038709*Nf + 68.66467304*Nf*Nf - 1.286400915*f1g11*Nf*Nf )
)

WCRV[{RV,G,2} ,10] = (
-0.5861279461*as4p*Nf - 24.76678064*as4p^2*Nf
+as4p^3*( -1201.206903*Nf + 72.23614791*Nf*Nf - 1.094394334*f1g11*Nf*Nf )
)

WCRV[{RV,G,2} ,12] = (
-0.5358430591*as4p*Nf - 25.51669345*as4p^2*Nf
+as4p^3*( -1351.047836*Nf + 73.7936445*Nf*Nf - 0.9344248731*f1g11*Nf*Nf )
)

WCRV[{RV,Q,L} ,2] = (
1.777777778*as4p + as4p^2*( 56.75530152 - 4.543209877*Nf - 3.950617284*f102*Nf )
+as4p^3*( 2544.598087 - 421.6908885*Nf - 7.736698288*f111*Nf + 11.8957476*Nf*Nf
+f102*( -213.9253076*Nf + 17.91326528*Nf*Nf ) )
)

WCRV[{RV,Q,L} ,4] = (
1.066666667*as4p + as4p^2*( 47.99398931 - 3.413333333*Nf - 0.6945185185*f102*Nf )
+as4p^3*( 2523.73902 - 383.0520013*Nf - 5.058869512*f111*Nf + 10.88895473*Nf*Nf
+f102*( -55.5530456*Nf + 2.348005487*Nf*Nf ) )
)

WCRV[{RV,Q,L} ,6] = (
0.7619047619*as4p + as4p^2*( 40.9961976 - 2.69569161*Nf - 0.2524824533*f102*Nf )
+as4p^3*( 2368.193775 - 340.0691069*Nf - 3.705612526*f111*Nf + 9.472190428*Nf*Nf
+f102*( -24.01322539*Nf + 0.7652692585*Nf*Nf ) )
)

WCRV[{RV,Q,L} ,8] = (
0.5925925926*as4p + as4p^2*( 35.87664406 - 2.231471683*Nf - 0.1217397796*f102*Nf )
+as4p^3*( 2215.210875 - 305.4730329*Nf - 2.913702563*f111*Nf + 8.337149534*Nf*Nf
+f102*( -12.97185267*Nf + 0.344362391*Nf*Nf ) )
)

WCRV[{RV,Q,L} ,10] = (
0.4848484848*as4p + as4p^2*( 32.01765947 - 1.908598248*Nf - 0.06856377261*f102*Nf )
+as4p^3*( 2081.213222 - 278.0172177*Nf - 2.397641695*f111*Nf + 7.452505612*Nf*Nf
+f102*( -7.947555677*Nf + 0.1841598535*Nf*Nf ) )
)

WCRV[{RV,Q,L} ,12] = (
0.4102564103*as4p + as4p^2*( 29.0058065 - 1.671007435*Nf - 0.04265241396*f102*Nf )
+as4p^3*( 1965.791047 - 255.8431044*Nf - 2.035689631*f111*Nf + 6.751061503*Nf*Nf
+f102*( -5.284573837*Nf + 0.1098463658*Nf*Nf ) )
)

WCRV[{RV,NS,L} ,14] = (
0.3555555556*as4p + as4p^2*( 26.5848844 - 1.488624298*Nf )
+as4p^3*( 1866.009187 - 237.5642566*Nf - 1.768138102*f111*Nf + 6.182499654*Nf*Nf )
)

WCRV[{RV,G,L} ,2] = (
0.666666667*as4p*Nf + 12.94776709*as4p^2*Nf
+as4p^3*( 407.280632*Nf - 20.23959748*Nf*Nf - 0.388939664*f1g11*Nf*Nf )
)

```

```

WCRV[{RV,G,L},4] = (
0.2666666667*as4p*Nf + 13.81659259*as4p^2*Nf
+as4p^3*( 767.7125421*Nf - 36.78419232*Nf*Nf - 0.3984298404*flg11*Nf*Nf )
)

WCRV[{RV,G,L},6] = (
0.1428571429*as4p*Nf + 10.26095364*as4p^2*Nf
+as4p^3*( 694.5092121*Nf - 27.9895081*Nf*Nf - 0.3055276256*flg11*Nf*Nf )
)

WCRV[{RV,G,L},8] = (
0.0888888889*as4p*Nf + 7.733545104*as4p^2*Nf
+as4p^3*( 592.3307972*Nf - 21.30333681*Nf*Nf - 0.2322211886*flg11*Nf*Nf )
)

WCRV[{RV,G,L},10] = (
0.06060606061*as4p*Nf + 6.023053074*as4p^2*Nf
+as4p^3*( 504.5424832*Nf - 16.70544537*Nf*Nf - 0.1805482229*flg11*Nf*Nf )
)

WCRV[{RV,G,L},12] = (
0.04395604396*as4p*Nf + 4.830676204*as4p^2*Nf
+as4p^3*( 433.9050534*Nf - 13.47418408*Nf*Nf - 0.1439099345*flg11*Nf*Nf )
)

(* The numerical values of the anomalous dimensions for the odd moments of $F_3$ *)

WgammaRV[{RV,NS}, 1] = 0

WgammaRV[{RV,NS}, 3] = (
5.555555556*as4p + as4p^2*( 70.88477366 - 5.12345679*Nf )
+as4p^3*( 1244.913602 - 196.4738081*Nf - 1.762002743*Nf*Nf )
)

WgammaRV[{RV,NS}, 5] = (
8.088888889*as4p + as4p^2*( 98.19940741 - 7.68691358*Nf )
+as4p^3*( 1720.942172 - 278.1581739*Nf - 2.366211248*Nf*Nf )
)

WgammaRV[{RV,NS}, 7] = (
9.780952381*as4p + as4p^2*( 116.4158903 - 9.437457798*Nf )
+as4p^3*( 2036.492478 - 330.8816595*Nf - 2.739358023*Nf*Nf )
)

WgammaRV[{RV,NS}, 9] = (
11.05820106*as4p + as4p^2*( 130.3414045 - 10.77682428*Nf )
+as4p^3*( 2277.19805 - 370.5905277*Nf - 3.006446616*Nf*Nf )
)

WgammaRV[{RV,NS}, 11] = (
12.08581049*as4p + as4p^2*( 141.6907901 - 11.86441897*Nf )
+as4p^3*( 2473.311857 - 402.691565*Nf - 3.212753995*Nf*Nf )
)

WgammaRV[{RV,NS}, 13] = (
12.94606135*as4p + as4p^2*( 151.2989044 - 12.78102552*Nf )
+as4p^3*( 2639.409887 - 429.7314605*Nf - 3.37996738*Nf*Nf )
)

(* and the coefficient functions are *)

```

```

WCRV[{RV,NS,3}, 1] = (
1 - 4*as4p + as4p^2*( -73.33333333 + 5.333333333*NF )
+as4p^3*( -2652.154437 + 513.3100408*NF - 11.35802469*NF*NF )
)

WCRV[{RV,NS,3}, 3] = (
1 + 1.666666667*as4p + as4p^2*( 14.25404015 - 6.742283951*NF )
+as4p^3*( -839.7638717 - 45.09953407*NF + 1.747689309*NF*NF )
)

WCRV[{RV,NS,3}, 5] = (
1 + 7.748148148*as4p + as4p^2*( 173.000629 - 19.39801646*NF )
+as4p^3*( 4341.081057 - 961.2756356*NF + 22.24125078*NF*NF )
)

WCRV[{RV,NS,3}, 7] = (
1 + 12.72248677*as4p + as4p^2*( 345.9910777 - 30.52332666*NF )
+as4p^3*( 11119.00053 - 1960.237096*NF + 43.10377964*NF*NF )
)

WCRV[{RV,NS,3}, 9] = (
1 + 16.9152381*as4p + as4p^2*( 520.0059615 - 40.35464229*NF )
+as4p^3*( 18771.99642 - 2975.924131*NF + 63.17127673*NF*NF )
)

WCRV[{RV,NS,3}, 11] = (
1 + 20.54831329*as4p + as4p^2*( 690.8719666 - 49.17096968*NF )
+as4p^3*( 26941.47987 - 3984.411605*NF + 82.24581704*NF*NF )
)

WCRV[{RV,NS,3}, 13] = (
1 + 23.76237745*as4p + as4p^2*( 857.1778817 - 57.18099124*NF )
+as4p^3*( 35426.82868 - 4976.080869*NF + 100.3509187*NF*NF )
);

(* Rule fl02 for selecting non-singlet or singlet, according to table in Sect. 5 *)
(* of Retey and Vermaseren hep-ph/0007294 *)
(* FIXME: For now, we don't specify rules fl(g)11 relevant only for NNNLO c^(3) *)
(* Rules fl2 and flg2 are not used at all *)

nsrules = {fl02->0, fl11->"[fl11-rule]", flg11->"[flg11-rule]"}
sirules = {fl02->1, fl11->"[fl11-rule]", flg11->"[flg11-rule]"}

(* Now we define functions that are exported to user *)
(* Since we don't need NNNLO astrong^3 c^(3), I'll put them to zero now *)

(* gamma non-singlet *)
Wgamma[{RV}, {V, NSM}, 1] = 0 (* according to paper *)
Wgamma[{RV}, {V, NSP}, n_?EvenQ /; 2<=n<=12] := WgammaRV[{RV,QQ}, n] /. nsrules
Wgamma[{RV}, {V, NSP}, 14] = WgammaRV[{RV,NS}, 14]
Wgamma[{RV}, {V, NSM}, n_?OddQ /; 1<=n<=13] := WgammaRV[{RV,NS}, n]

(* gamma singlet *)
Wgamma[{RV}, {V, OP_}, n_?EvenQ /; 2<=n<=12] := WgammaRV[{RV,OP}, n] /. sirules

(* Coeff. of F2 *)
(* non-singlet*)
Wc[{RV}, {V,F2,NSP}, n_?EvenQ /; 2<=n<=12] := WCRV[{RV,Q,2}, n] /. nsrules /.as4p^3->0
Wc[{RV}, {V,F2,NSP}, 14] = WCRV[{RV,NS,2}, 14] /. nsrules /.as4p^3->0
(* singlet*)
Wc[{RV}, {V,F2,OP_}, n_?EvenQ /; 2<=n<=12] := WCRV[{RV,OP,2}, n] /. sirules /.as4p^3->0

(* Coeff. of FL *)
(* non-singlet*)

```



```

Wc[{RV}, {V,FL,NSP}, n_?EvenQ /; 2<=n<=12] := WCRV[{RV,Q,L}, n] /. nsrules /.as4p^3->0
Wc[{RV}, {V,FL,NSP}, 14] = WCRV[{RV,NS,L}, 14] /. nsrules /.as4p^3->0
(* singlet*)
Wc[{RV}, {V,FL,OP_}, n_?EvenQ /; 2<=n<=12] := WCRV[{RV,OP,L}, n] /. sirules /.as4p^3->0

(* Coeff. of F3 *)
(* non-singlet *)
Wc[{RV}, {V,F3,NSM}, n_?OddQ /; 1<=n<=13] := WCRV[{RV,NS,3}, n] /.as4p^3->0

(* Coeff. of F1: it's just F2-FL *)
(* FIXME: There is no checking of argument n here *)
Wc[{RV}, {V,F1,OP_}, n_] := Wc[{RV}, {V,F2,OP}, n] - Wc[{RV}, {V,FL,OP}, n]

(* Extracting particular coefficient of alpha^p *)
(* FIXME: There is no checking of argument n here *)

Wgamma[{RV}, {plr_,OP_}, p_?IntegerQ /; 0<=p<=2, n_] := Coefficient[Wgamma[{RV}, {plr,OP}, n], as4p^(p+1)]
Wc[{RV}, {plr_,ff_ /; MemberQ[{F1,F2,F3}, ff], OP_ /; MemberQ[{Q,NS}, OP]}, 0, n_] := 1
Wc[{_,_}, 0, n_] := 0
Wc[{RV}, {plr_,ff_,OP_}, p_?IntegerQ /; 1<=p<=2, n_] := Coefficient[Wc[{RV}, {plr,ff,OP}, n], as4p^p]

```

## References

- [1] W. L. van Neerven and A. Vogt, *NNLO evolution of deep-inelastic structure functions: The non-singlet case*, *Nucl. Phys.* **B568** (2000) 263–286 [[hep-ph/9907472](#)].
- [2] M. Gluck, E. Reya, M. Stratmann and W. Vogelsang, *Next-to-leading order radiative parton model analysis of polarized deep inelastic lepton - nucleon scattering*, *Phys. Rev.* **D53** (1996) 4775–4786 [[hep-ph/9508347](#)].
- [3] J. Blumlein and S. Kurth, *On the mellin transform of the coefficient functions of  $f_l(x, q^{*2})$* , [hep-ph/9708388](#).
- [4] A. J. Buras, *Asymptotic freedom in deep inelastic processes in the leading order and beyond*, *Rev. Mod. Phys.* **52** (1980) 199.
- [5] G. Curci, W. Furmanski and R. Petronzio, *Evolution of parton densities beyond leading order: The nonsinglet case*, *Nucl. Phys.* **B175** (1980) 27.
- [6] S. Moch, J. A. M. Vermaseren and A. Vogt, *The three-loop splitting functions in qcd: The non-singlet case*, *Nucl. Phys.* **B688** (2004) 101–134 [[hep-ph/0403192](#)].
- [7] J. Blumlein and S. Kurth, *Harmonic sums and mellin transforms up to two-loop order*, *Phys. Rev.* **D60** (1999) 014018 [[hep-ph/9810241](#)].
- [8] E. G. Floratos, C. Kounnas and R. Lacaze, *Higher order qcd effects in inclusive annihilation and deep inelastic scattering*, *Nucl. Phys.* **B192** (1981) 417.

- [9] A. Vogt, S. Moch and J. A. M. Vermaseren, *The three-loop splitting functions in qcd: The singlet case*, *Nucl. Phys.* **B691** (2004) 129–181 [[hep-ph/0404111](#)].
- [10] W. A. Bardeen, A. J. Buras, D. W. Duke and T. Muta, *Deep inelastic scattering beyond the leading order in asymptotically free gauge theories*, *Phys. Rev.* **D18** (1978) 3998.
- [11] A. Retey and J. A. M. Vermaseren, *Some higher moments of deep inelastic structure functions at next-to-next-to leading order of perturbative qcd*, *Nucl. Phys.* **B604** (2001) 281–311 [[hep-ph/0007294](#)].
- [12] W. L. van Neerven and A. Vogt, *NNLO evolution of deep-inelastic structure functions: The singlet case*, *Nucl. Phys.* **B588** (2000) 345–373 [[hep-ph/0006154](#)].
- [13] E. B. Zijlstra and W. L. van Neerven, *Order  $\alpha(s)^{**2}$  corrections to the polarized structure function  $g_1(x, q^{**2})$* , *Nucl. Phys.* **B417** (1994) 61–100.